

CAE of Sensors and Actuators

Electrostatic Field Example

In this example, we would like to simulate the electric field and the capacitance of a **cylindrical** capacitor, including the surrounding air. The inner electrode consists of a centered rod and the outer electrode is the metallic coat. Between inner and outer electrode a voltage of 1 V shall be applied.

The dimensions are as follows

- Radius of inner electrode = 1 mm
- Radius of outer electrode = 6 mm
- Thickness of outer electrode = 1 mm
- Height of cylinder = 100 mm
- Size of additional boundary = 10 mm




Figure 1: Domain regions of the capacitor model

This example is divided into three parts:

- Part I. Meshing with Ansys
- Part II. Simulation with NACS
- Part III. Postprocessing with Gid

Part I. Meshing with Ansys

Tutorial: October 20th, 2014

Further information:

- `ansysCommandReference.pdf` - list of all usable Ansys commands (without special NACS commands)
- `ansysBasics.pdf` - overview over basic steps

Creating an Ansys input script

- **Preparations**

1. make sure that you have a file called **start145.ans** in your home directory (this file defines some macros needed for correct mesh creation)
2. create an empty script file in your simulation directory (e.g. `ansysTutorial.in`)
3. add the following header to the beginning of your script file (! marks a comment, comments need not to be copied):

```
!!!! HEADER !!!!
FINI      !Exits all old scripts
/CLEAR    !Clears workspace
/PREP7    !Starts command interface
NACSINIT  !Starts NACS interface
```

- **Define parameter**

In Ansys scripts you can easily define parameter by assigning a value to a name via `=`. Using parameter is an elegant way to keep your mesh script flexible and reusable. For the given geometry add the following parameter to your script file (all parameter have to be in SI units):

```
!!!! PARAMETER !!!!
r_inner = 1e-3      !Radius of inner electrode in m
r_outer = 6e-3      !Radius of outer electrode in m
t_outer = 1e-3      !Thickness of outer electrode in m
h_cylinder = 100e-3 !Height of cylinder in m
s_air = 10e-3       !Size of additional air boundary in m
dx = 1e-3           !Meshsize (the smaller the mesh size,
                    !the more accurate the results, BUT
                    !the more expensive the simulation!)
```

- **Create Geometry**

Since the geometry is axisymmetric only half of the cross section has to be modeled. In Ansys there exist two approaches on how to setup the geometry.

1. bottom-up approach
In the bottom-up approach you start by creating the corner points of your geometry (called Keypoints) and connect them manually to lines, areas and volumes.
2. top-down approach
In the top-down approach you start by creating volumes or areas and define subvolumes and subareas by using cutting or overlapping commands (e.g. `AINP` or `AOVLAP` → see `ansysCommandReference.pdf` for further information).

In general both approaches lead to the same result if used correctly. The correct usage, especially of the overlapping commands, requires some practice and is not easy to master. The geometry for this model example as well as the geometries of the exercise tasks, however, can be modeled without the usage of those complex commands. The geometry for this model example can e.g. be created out of six simple rectangles (using the RECTNG command):

```
!!!! CREATE GEOMETRY !!!!
! RECTNG,x_min,x_max,y_min,y_max -> rectangle from x_min,y_min to x_max,y_max
!Electrodes
RECTNG,0,r_inner,0,h_cylinder
RECTNG,r_outer,r_outer+t_outer,0,h_cylinder
!Dielectricum
RECTNG,r_inner,r_outer,0,h_cylinder
!Surrounding air consisting of three rectangles
RECTNG,r_outer+t_outer,r_outer+t_outer+s_air,0,h_cylinder
RECTNG,0,r_outer+t_outer+s_air,-s_air,0
RECTNG,0,r_outer+t_outer+s_air,h_cylinder,h_cylinder+s_air
```

- **Connect geometry**

A very important step in the meshing process is the connection of single geometric entities. If this is not done the later mesh file could see two adjacent areas as not connected. In the best case the later simulation will see an invisible wall there, in the worst case the simulation will crash. A good way to connect your geometry is to use the Ansys gluing commands LGLUE, AGLUE, VGLUE which will connect lines, areas or volumes. In general it is sufficient to call the command of highest dimensionality (e.g. if you have volumes a VGLUE command should also connect the areas and the lines of the volumes correctly). For the given example the connection step could look like:

```
!!!! CONNECT GEOMETRY !!!!
ASEL,all      !Select all areas
AGLUE,all     !Glue all selected areas
```

- **Create mesh**

After gluing all entities together the actual meshing of the geometry is done. For the meshing process one has to set a meshsize to lines (LESIZE) or areas (AESIZE) at first. Then one can mesh lines by using LMESH, areas by AMESH and volumes by VMESH. For this example as well as for the assignments only the AMESH command is needed (due to the NACS interface all needed lines will be meshed automatically).

```
!!!! SETUP MESH !!!!
! 1. set meshsize (for our purpose usually on lines)
LSEL,all      !Select all lines
LESIZE,all,dx !Set the meshsize dx on all selected lines

! 2. mesh all areas
SETELEMS,'quadr','' !Use elements of first order
                        !(for later mechanic simulations you will
                        !also need second order elements using
                        !'quadr','quad' instead of 'quadr','')
ASEL,all      !Select all areas
AMESH,all     !Mesh all selected areas
```

- **Define components**

Until now the modeled geometry consists of several rectangles, lines and points (also called nodes), but yet the information is missing which geometric entities shall later be part of air,

dielectricum or one of the electrodes. To specify this information we have to create components (the creation of components is essential for the later saving process where the mesh is written to file). To create a component one has to select the desired entity (NSEL for nodes, LSEL for lines, ASEL for areas or VSEL for volumes) and use the CM command.

```
!!!! CREATE COMPONENTS !!!!!
! CM,name,etype -> creates a component called 'name' of all currently selected
! entities of type 'etype' (can be node, line, area, volume)
ASEL,s,loc,x,0,r_inner      !Select ('s' for select) all areas with x coordinates
                             !between 0 and r_inner;
                             !Note that 's' unselects all previously selected areas;
                             !To add areas to the current selection use 'a'
ASEL,r,loc,y,0,h_cylinder   !Reselect ('r' for reselect) from the current selection
                             !all areas with y between 0 and h_cylinder
CM,hotEL,area               !Put all currently selected areas to component hotEL

ASEL,s,loc,x,r_inner,r_outer
ASEL,r,loc,y,0,h_cylinder
CM,diel,area

ASEL,s,loc,x,r_outer,r_outer+t_outer
ASEL,r,loc,y,0,h_cylinder
CM,gndEL,area

!To select all air regions one can select all three rectangles by their coordinates
!or one can select all areas and unselect the already named components
ASEL,all                    !Select all areas
ASEL,u,,hotEL               !Unselect component named hotEL from current selection
                             !Note that the spaces for loc and x/y where kept free

ASEL,u,,gndEL
ASEL,u,,diel                !Now only the three air areas are left
CM,air,area

!Select the top line and the observer point
!Note that only existing lines and nodes can be selected (e.g. you cannot select
!a line from the top left to the bottom right corner here)
tol = dx/5                  !When selecting lines or nodes allow a little tolerance
                             !as otherwise the selection might fail.
                             !Setting tol to a fraction of the meshsize is in most cases
                             !a good idea.
LSEL,s,loc,x,0,r_outer+t_outer+s_air
LSEL,r,loc,y,h_cylinder+s_air-tol,h_cylinder+s_air+tol
CM,topLine,line

NSEL,s,loc,x,r_outer+t_outer+s_air-tol,r_outer+t_outer+s_air+tol
NSEL,r,loc,y,h_cylinder-tol,h_cylinder+tol
CM,observer,node
```

- **Write out mesh**

The last step in each mesh script is to write out the created elements to a mesh file. In comparison to standard Ansys meshing the NACS meshing command `WRITEMODEL` is used, which adds some extra functionality and creates the correct file format. This command, however, requires the creation of **regions** (for areas and volumes) and **groups** (for lines and nodes) for all entities which shall later be selectable in the mesh or which are needed by other entities (e.g. if you do not use the top line later you do not need to write it out, but you cannot write out the top line without writing out the surrounding air, as the line is part of the air region).

To create regions and groups do the following:

```
!!!! WRITE OUT MESH !!!!
! 1. create regions for all area- and volume-components
! REGION,'regionName','compName' -> creates region called regionName
!                                     !of component compName
REGION,'hotEL_region','hotEL'
REGION,'gndEL_region','gndEL'
REGION,'diel_region','diel'
REGION,'air_region','air'
! 2. create groups for all line- and node-components
! GROUP,'groupName','compName' -> creates group called groupName
!                                     !of component compName
GROUP,'topLine_group','topLine'
GROUP,'observer_group','observer'
! 3. write out model
WRITEMODEL,'ansysTutorial'
```

Creating a mesh file from script

To create a mesh file from the previously created mesh script you have three possibilities:

1. pass mesh script to Ansys via command line:
 ansys-batch -i scriptName
 → fast, but no visualization of the mesh
2. open Ansys via **ansys**, then select File -> Read Input from ... -> scriptName
 → mesh and geometry visible
3. open Ansys via **ansys**, then copy and paste the content of your mesh script into the command line and press Enter
 → slow, but geometry and mesh can be created step by step

Check and view your mesh

To view your mesh you have to create it in an open ansys window first (see option 2. or 3. from the previous section). Then select the type of entity you want to investigate by using the selection commands from above in the ansys command line, e.g.

```
nsel,all                !Select all nodes
cmselect,s,air           !Select component named air
asel,s,,,air            !Select area-component named air
lsla,all                !Select all lines of the currently selected area
lselect,r,loc,x,r_inner,r_outer !Reselect all lines with x between r_inner and r_outer
eset,all                !Select all elements
```

The selected entities can then be viewed by using NPLLOT for nodes, LPLLOT for lines, APLLOT for areas, CMPLOT for components and finally EPLLOT for elements. Under PlotCtrls -> Numbering you can also switch on the node, line, area, ... numbers.